

## 4 Non-Linear Oscillations and Chaos

### 4.1 Non-Linear Differential Equations

Up to now, we have considered differential equations with terms that are proportional to the acceleration, the velocity, and the position:

$$m\ddot{x} + f(\dot{x}) + g(x) = h(t) \quad (4.1)$$

where  $f(\dot{x})$  is proportional to  $\dot{x}$  and  $g(x)$  is proportional to  $x$ . Such an equation is called a **linear differential equation**. Although many important applications can be described in terms of a linear differential equation, many other applications require the inclusion of terms that are non linear, e.g.  $\dot{x}^2$  and  $x^3$ . In those cases, the system is called **non-linear**. Differential equations can become non-linear if for example:

1. The restoring force is not proportional to the displacement  $x$ .
2. The damping force is not proportional to the velocity  $v$ .

An example of the first case is the pendulum. The pendulum will exhibit simple-harmonic motion if the angular displacement is small. In that case, the restoring force, which is proportional to  $\sin\theta$ , is approximately proportional to  $\theta$ . However, if the angular displacement is large,  $\sin\theta$  is no longer proportional to  $\theta$ , and the system will no longer exhibit simple harmonic motion. Systems in which the restoring force is non-linear can be divided in to two groups:

#### 1. Systems for which the force is symmetric around the equilibrium position.

Examples of such forces are shown in Fig. 4.1. The non-linear behavior, which is added to the linear behavior, must be proportional to e.g.  $x^3$  or  $x^5$  since terms proportional to  $x^2$  would reduce the magnitude of the restoring force on one side, while increasing its magnitude on the opposite side:

$$F(x) = -kx + \varepsilon x^3 \quad (4.2)$$

If  $\varepsilon > 0$ , the magnitude of the force is decreased compared to the linear situation. Such a system is called a **soft system**. If  $\varepsilon < 0$ , the magnitude of the force is increased compared to the linear situation. Such a system is called a **hard system**.

#### 2. Systems for which the force is asymmetric around the equilibrium position.

Examples of such forces are shown in Fig. 4.2. The non-linear behavior, which is added to the linear behavior, must be proportional to e.g.  $x^2$  or  $x^4$  since terms proportional to  $x^3$  would reduce the magnitude of the restoring force on both sides:

$$F(x) = -kx + \lambda x^2 \quad (4.3)$$

If  $\lambda > 0$ , the force is increased compared to the linear situation. If  $\lambda < 0$ , the force is decreased compared to the linear situation. In both cases, the system is hard on one side of the equilibrium position, and soft on the opposing side.

### 4.2 Phase Diagrams for Non-Linear Systems

If the potential energy of the linear system is known, it is easy to construct the phase diagram. The procedure used to construct the phase diagram relies on conservation of energy:

$$E = U(x) + \frac{1}{2}m\dot{x}^2 \quad (4.4)$$

or

$$\dot{x} = \frac{2}{m} \sqrt{E - U(x)} \quad (4.5)$$




Figure 4.1: Examples of forces that are symmetric around the equilibrium position and their corresponding potential energy distributions.




Figure 4.2: An example of a force that is asymmetric around the equilibrium position and its corresponding potential energy distribution.




Figure 4.3: (a) Potential energy distributions around stable (left) and unstable (right) equilibrium positions. (b) Phase diagrams for stable (left) and unstable (right) equilibrium positions.

Examples of phase diagrams for stable and unstable equilibrium positions are shown in Fig. 4.3. The curves of constant energy are closed when we are dealing with a stable equilibrium, while they are not closed when we are dealing with unstable equilibrium.

### 4.3 The van der Pol Equation

An important type of non-linear differential equation was used to describe the non-linear oscillations in electronic circuits containing vacuum tubes. The van der Pol differential equation has the following form:

$$\ddot{x} + \mu(x^2 - a^2)\dot{x} + \omega_0^2 x = 0 \quad (4.6)$$

This equation is non-linear due to the presence of the  $x^2$  term. One can interpret this equation as the equation of a system that has variable damping. In addition, the “damping” coefficient can be positive or negative, depending on the value of  $a$ . Based on what we have seen for damping in Chapter 3, we expect the following behavior for the solution of this differential equation:

1. If  $|x| > |a|$ , the “damping” coefficient is positive and the amplitude of the motion decreases.
2. If  $|x| < |a|$ , the “damping” coefficient is negative and the amplitude of the motion increases.

After a period of time we can imagine that the “damping” coefficient approaches zero and simple harmonic motion, with constant amplitude, will result. The curve mapped out in phase space in this case is called the **limit cycle**, and eventually no matter what the starting point, the system will follow the limit cycle.

To study the time evolution of the van der Pol equation, we need to use numerical methods to solve the differential equation. I will illustrate the use of Mathematica to solve this equation. The following code will allow you to play with the solutions of the van der Pol equation (the code can be found in the file *vanDerPolSolution* in the Mathematica folder under Computer Code on our website):

```
(* Set the values of the various parameters *)
mu = 0.05;
a = 1.0;
w0 = 1;

(* Solve the differential equations with the given set of initial \
conditions. *)
sol = NDSolve[
  {x'[t] == v[t],
   v'[t] == -mu*(x[t]*x[t] - a*a)*v[t] - w0*w0*x[t],
   x[0] == 1,
   v[0] == 0},
  {x, v}, {t, 0, 50*2*Pi}, MaxSteps -> 20000];

(* Plot the solution of the differential equations *)
ParametricPlot[Evaluate[{x[t], v[t]} /. sol], {t, 0, 50*2*Pi},
  PlotRange -> All,
  AxesLabel -> {"x (rad)", "v (rad/s)"}]
```

The results of this type of calculations, for different starting parameters and damping terms, are shown in Figs. 4.4 and 4.5. Fig. 4.4 shows that the limit cycle is the same for different initial values of  $x_0$ . Looking at Fig. 4.5, we observe that the shape of the limit cycle is a function of the damping parameter. Figure 4.6 shows that an increase of the damping parameter decreases the time required to approach the limit cycle.

#### 4.4 The Plane Pendulum

A good example of a system with non-linear behavior is the plane pendulum. In many introductory physics courses, the simple pendulum is used to introduce harmonic motion. However, the simple pendulum will only exhibit simple harmonic motion if the angular displacement is small (small enough to ensure that  $\sin\theta$  is approximately  $\theta$ ). If we can not make the assumption that the angular displacements are small, the simple pendulum will show non-linear behavior. The differential equation describing the system is given by

$$\ddot{\theta} + \omega_0^2 \sin\theta = 0 \quad (4.7)$$

where

$$\omega_0^2 = \frac{g}{l} \quad (4.8)$$

and  $l$  is the length of the pendulum.

Before we look in detail at the phase diagrams for this pendulum, we should first consider what we expect to see. Let's assume that at time  $t = 0$  s, the pendulum is located at its equilibrium position, and it has a linear velocity  $v_0$  at that time.




Figure 4.4: Solution of the van der Pol equation with  $\mu = 0.05$ ,  $a = 1$  rad,  $\omega_0 = 1$  rad/s,  $x_0 = 4$  rad,  $v_0 = 0$  rad/s (left) and  $\mu = 0.05$ ,  $a = 1$  rad,  $\omega_0 = 1$  rad/s,  $x_0 = 1$  rad,  $v_0 = 0$  rad/s (right).




Figure 4.5: Solution of the van der Pol equation with  $\mu = 0.05$ ,  $a = 1$  rad,  $\omega_0 = 1$  rad/s,  $x_0 = 2$  rad,  $v_0 = 0$  rad/s (left) and  $\mu = 0.5$ ,  $a = 1$  rad,  $\omega_0 = 1$  rad/s,  $x_0 = 2$  rad,  $v_0 = 0$  rad/s (right).




Figure 4.6: Solution of the van der Pol equation with  $\mu = 0.05$ ,  $a = 1$  rad,  $\omega_0 = 1$  rad/s,  $x_0 = 4$  rad,  $v_0 = 0$  rad/s (left) and  $\mu = 0.5$ ,  $a = 1$  rad,  $\omega_0 = 1$  rad/s,  $x_0 = 4$  rad,  $v_0 = 0$  rad/s (right).

1. If the velocity at  $t = 0$  is small, the system will oscillate back and forth between a minimum and maximum value of the angular displacements, and the phase diagram is a closed curve (see Fig. 4.7).
2. If the velocity at  $t = 0$  is sufficiently large, the system will continue to rotate in one direction, and the angular velocity will show an oscillatory behavior (but without a sign reversal). An example of this response is shown in Fig. 4.8.
3. The limiting case will be the case where the velocity at  $t = 0$  is such that the pendulum makes it just to its highest possible position ( $y = 2l$ ). Assuming the mass of the pendulum is connected via a weightless rod to the pivot point, the total energy of the mass will have to be  $2mgl$  (kinetic energy will be equal to 0 at the top position). The kinetic energy at the equilibrium must therefore be  $2mgl$ , and the linear velocity at that position will be equal to  $2\sqrt{gl}$ . The angular velocity at the equilibrium position will thus be equal to  $(2\sqrt{gl}/l) = 2\sqrt{g/l} = 2\omega_0$ . For the phase diagrams shown in Figs. 4.7 and 4.8, we have assumed  $\omega_0 = 1$  rad/s, and the “critical” angular velocity is thus equal to 2 rad/s.

Figures 4.7 and 4.8 were made using Mathematica and can be reproduced using the following code, which is also contained in Mathematica notebook *planePendulum* that is available from our website:

```
(* Set the values of the various parameters *)
w0 = 1;

(* Solve the differential equations with the given set of initial \
conditions. *)
sol = NDSolve[
  {x'[t] == v[t],
   v'[t] == -w0*w0*Sin[x[t]],
   x[0] == 0,
   v[0] == 1.5},
  {x, v}, {t, 0, 50*2*Pi}, MaxSteps -> 20000];

(* Plot the solution of the differential equations *)
ParametricPlot[Evaluate[{x[t], v[t]} /. sol], {t, 0, 5*2*Pi},
  AxesLabel -> {"x (rad)", "v (rad/s)"}]
```




Figure 4.7: Solution of the differential equation associated with the plane pendulum with  $\omega_0 = 1$  rad/s,  $x_0 = 0$  rad,  $v_0 = 1$  rad/s (left) and  $\omega_0 = 1$  rad/s,  $x_0 = 0$  rad,  $v_0 = 1.5$  rad/s (right). **Note that the velocity  $v_0$  is below the critical velocity.**

Another way to look at the phase diagram is to use contour plots. Rather than solving the differential equation for a specific set of initial parameters, and thus a specific energy, the contour plot provides an overview of the solution for a range of initial conditions. Consider a point on the phase diagram corresponding to an angular position  $\theta$  and an angular velocity  $\omega$ . The kinetic and potential energies at this location are:

$$T = \frac{1}{2}I\omega^2 = \frac{1}{2}ml^2\dot{\theta}^2 \quad (4.9)$$

and

$$U = mgl(1 - \cos\theta) \quad (4.10)$$

The total energy  $E$  is thus equal to

$$E = T + U = \frac{1}{2}ml^2\dot{\theta}^2 + mgl(1 - \cos\theta) \quad (4.11)$$

A contour plot, showing the total energy  $E$  as function of  $\theta$  and  $\omega$ , shows the nature of the motion that will occur when we give specific initial conditions for the pendulum. An example of a contour plot, obtained for  $m = 1$  kg and  $l = 1$  m is shown in Fig. 4.9. This contour plot was generated using the following Mathematica code, which is also contained in Mathematica notebook *planePendulumContours* that is available from our website:

```
l = 1;
g = 9.8;
ContourPlot[(1./2.)*l^2*w^2 + g*l*(1. - Cos[theta]),
  {theta, -4 Pi, 4 Pi}, {w, -15, 15},
  PlotLegends -> Automatic,
  FrameLabel -> {"theta (rad)", "omega (rad/s)"}]
```




Figure 4.8: Solution of the differential equation associated with the plane pendulum with  $\omega_0 = 1 \text{ rad/s}$ ,  $x_0 = 0 \text{ rad}$ ,  $v_0 = 2 \text{ rad/s}$  (left) and  $\omega_0 = 1 \text{ rad/s}$ ,  $x_0 = 0 \text{ rad}$ ,  $v_0 = 2.5 \text{ rad/s}$  (right). **Note that the critical velocity  $v_0$  is 2 rad/s.**




Figure 4.9: Contour plot for a plane pendulum ( $m = 1 \text{ kg}$ ,  $l = 1 \text{ m}$ ).




Figure 4.10: A damped pendulum, driven about its pivot point.

## 4.5 Chaos

Important principles of chaos can be studied using a simple driven pendulum. Consider the pendulum shown in Fig. 4.10, which is driven about its pivot point. We assume that the pendulum is damped with a damping coefficient  $b$ . The magnitude of the total torque around the pivot point is equal to

$$\tau = I\ddot{\theta} = ml^2\ddot{\theta} \quad (4.12)$$

The total torque is equal to the sum of the torques due to the gravitational force, the damping force, and the driving force:

$$\tau = -b\dot{\theta} - mgl\sin\theta + \tau_d\cos(\omega t) \quad (4.13)$$

The differential equation describing the motion of this driven pendulum is given by

$$\ddot{\theta} = -\frac{b}{ml^2}\dot{\theta} - \frac{g}{l}\sin\theta + \frac{\tau_d}{ml^2}\cos(\omega t) \quad (4.14)$$

This differential equation is a non-linear equation due to the  $\sin\theta$  term. This equation has the following general form:

$$\ddot{x} = -c\dot{x} - \sin x + F\cos(\omega t) \quad (4.15)$$

The solution to this equation can be studied using numerical methods, and we will focus on the results obtained with Mathematica. The code required to study the solutions of this equation can be found in the file *chaosPendulum* in the Mathematica folder under Computer Code on our website:

```
(* Set the values of the various parameters *)
c = 0.2;
w = 0.694;
F = 0.52;
(* Increase F in steps of 0.1 to see the different solutions in the chapter notes. *)
pi = N[Pi];
cycles = 50;
```

```

steps = 30;

(* Solve the differential equations with the given set of initial \
conditions. *)
sol = NDSolve[
  {x'[t] == v[t],
   v'[t] == -c*v[t] - Sin[x[t]] + F*Cos[w*t],
   x[0] == 0.8,
   v[0] == 0.8},
  {x, v}, {t, 0, (cycles*(2 Pi)/w)}, MaxSteps -> 20000];

(* Create a function "reduce" that translate all angles back to the \
region between -Pi and +Pi *)
reduce[x_] := Mod[x, 2 pi] /; Mod[x, 2 pi] <= pi;
reduce[x_] := (Mod[x, 2 pi] - 2 pi) /; Mod[x, 2 pi] > pi;

(* Create a table with the angles mapped to the -Pi to Pi region. *)
points = Flatten[
  Table[
    {x[t], v[t]} /. sol,
    {t, 0, (cycles*(2 Pi)/w), (1/steps) (2 Pi/w) // N}
  ], 1
];
xposition = Table[{i, 1}, {i, Length[points]}];
newpoints = MapAt[reduce, points, xposition];

(* Plot the solution of the differential equations: v[t] vs t. *)
ParametricPlot[
  Evaluate[{t, v[t]} /. sol], {t, 0, (cycles*(2 Pi)/w)},
  PlotRange -> {{{(cycles - 20)*(2 Pi)/w}, (cycles*(2 Pi)/w)}, {-3, 3}},
  AxesLabel -> {"t/t0", "v (rad/s)"},
  AspectRatio -> 1/2
]

(* Plot the phase diagram. *)
ParametricPlot[
  Evaluate[{x[t], v[t]} /.
    sol], {t, ((cycles - 20)*(2 Pi)/w), (cycles*(2 Pi)/w)},
  AxesLabel -> {"x (rad)", "v (rad/s)"},
  AspectRatio -> 1/2
]

(* Plot the phase diagram (using the reduced angle). *)
ListPlot[
  newpoints,
  PlotRange -> {{-3, 3}, {-3, 3}},
  AxesLabel -> {"x (rad)", "v (rad/s)"},
  AspectRatio -> 1/2
]

(* Make the poincare plot. *)
poincare =
  Table[newpoints[[n]], {n, 1 + 25 steps, Length[points], steps}];
Length[poincare];

```




Figure 4.11: Velocity versus time for the driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.52$ .

```
ListPlot[
  poincare,
  PlotRange -> {{-3, 3}, {-3, 3}},
  AxesLabel -> {"x (rad)", "v (rad/s)"},
  PlotStyle -> PointSize[0.015],
  AspectRatio -> 1/2
]
```

There are many different ways of looking at the results of this calculation. Consider the case where  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.52$ . A graph showing the velocity versus time is shown in Fig. 4.11. The period shown in Fig. 4.11 does not include the initial period when transient effects may dominate. The motion is periodic, but not simple harmonic. The average velocity is negative, indicating that on average the solution will move to more negative angles.

Another way to look at the solution is to construct the phase diagram. The phase diagram for the solution shown in Figure 11 is shown in Fig. 4.12. Figure 4.12(left) shows the velocity versus angle. As expected, the system will move towards more negative angles. Figure 4.12(right) shows the similar information to what is shown in Fig. 4.12(left), except that the angle is reduced to an angle between  $-\pi$  and  $+\pi$ .

The phase diagrams shown in Fig. 4.12 are often difficult to interpret. An approach that was developed to emphasize the possible periodic nature of the solution is the Poincare representation. In this representation, we plot the velocity versus position at times separated by one period ( $2\pi/\omega$ ). If the motion is periodic, each new point in the Poincare diagram will be plotted at the same position as the previous point: the plot shows thus a single point. **Note: the number of points shown will provide information about the periodicity; the actual position does not provide much information.** The Poincare representation of the phase diagrams shown in Fig. 4.12 is shown in Fig. 4.13. There are 3 points in the Poincare representation, suggesting a periodic solution (but not simple harmonic).

The solution of the driven pendulum is very sensitive to the parameters that are being used in the calculation. The transition of chaos is characterized by an increase in the number of points in the Poincare diagrams. Consider what happens if we change one of the parameters of our differential equation. Figure 4.14 shows the results of a calculation when we increase  $F$  from 0.52 to 0.62. The solution has become a simple harmonic, as can be seen by the fact that the Poincare plot contains only a single point.

Figures 4.15, 4.16, and 4.17 show the results of increasing the force  $F$  to 0.72, 0.82, and 0.92. The solution is chaotic when  $F = 0.72$  and  $F = 0.82$ , but the solution is periodic when  $F = 0.92$ .




Figure 4.12: Phase diagrams for the driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.52$  (left). The diagram shown on the right contains the same information as in a) except that the angles are mapped on the range between  $-\pi$  and  $+\pi$




Figure 4.13: Poincaré representation of the motion of a driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.52$ .




Figure 4.14: Solutions of the motion of a driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.62$ .




Figure 4.15: Solutions of the motion of a driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.72$ .




Figure 4.16: Solutions of the motion of a driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.82$ .




Figure 4.17: Solutions of the motion of a driven pendulum with  $c = 0.2$ ,  $\omega = 0.694$ , and  $F = 0.92$ .




Figure 4.18: Examples of the logistic equation map for two different values of  $\alpha$ .

## 4.6 Logistic Equations

Chaos can also develop when we work with maps. Maps are used to describe the evolution of a system, and can developed if we know how the future depends on the immediate past. For example, consider a situation where the value of a system parameter at time  $t_{N+1}$ ,  $x_{N+1}$ , depends on the value of this parameter at time  $t_N$ ,  $x_N$ :

$$x_{N+1} = \alpha x_n(1 - x_n) \quad (4.16)$$

The function describing this relation is called the **logistic equation**, and it depends on both  $x$  and on  $\alpha$ :

$$f(\alpha, x) = \alpha x(1 - x) \quad (4.17)$$

For certain values of  $\alpha$ , the system will evolve towards a stable equilibrium while for other starting value the system may evolve towards more than one possible solution. Two examples, for two different values of  $\alpha$ , are shown in Figure 16.

In the example shown in Fig. 4.18b, there are two solutions to the logistic map. For higher value of  $\alpha$  there may be more solutions. The best way to view the possible modes of evolution is to use a bifurcation diagram. In a bifurcation diagram, we plot the value of  $x_n$  as function of the parameter  $\alpha$ . We can construct a bifurcation diagram using numerical methods, and we will focus on the results obtained with Mathematica. The code required to study the solutions of this equation can be found in the file *bifurcation* in the Mathematica folder under Computer Code on our website. A bifurcation diagram for the logistic equation discussed in this section is shown in Fig. 4.19. The two diagrams in Fig. 4.19 show the bifurcation diagrams that are created for a starting value  $x_0 = 0.1$  (left) and  $x_0 = 0.9$  (right). We see that for values of  $\alpha$  below 3 there is only one solution. For values of  $\alpha$  between 3 and 3.45, there are two possible solutions.




Figure 4.19: Bifurcation diagrams for  $f(\alpha, x) = \alpha x(1 - x)$  for a starting value of  $x_0 = 0.1$  (left) and  $x_0 = 0.9$  (right).




Figure 4.20: Bifurcation diagrams for  $f(\alpha, x) = \alpha x(1 - x)$  for a starting value of  $x_0 = 0.0$  (left) and  $x_0 = 1.0$  (right).

For values between 3.45 and 3.55 there are four possible solutions. Above  $\alpha = 3.55$  the system has many possible solutions, and the system is said to be chaotic. However, as Fig. 4.19 shows, for very specific values of  $\alpha$  the system returns back to a highly-ordered state with a small number of solutions. As can be seen, the bifurcation diagram does not change when we change the initial conditions, except for the special cases shown in Fig. 4.20 where  $x_0 = 0.0$  (left) and  $x_0 = 1.0$  (right).

Changes to the logistic equation will change the bifurcation diagram. Consider for example the following logistic equation:

$$f(\alpha, x) = \alpha x(1 - x)^2 \quad (4.18)$$

The corresponding bifurcation diagram is shown in Fig. 4.21. Many of the features are similar, except that we see that the first doubling of the number of solutions does not occur until  $\alpha$  exceeds 4.

When the time evolution of a system depends sensitively on the initial conditions, the system is said to show the characteristics of chaos. In the bifurcation diagrams shown in Fig. 4.19, we see that a particular system can have non-chaotic behavior for a specific range of parameters and chaotic behavior for a different range of parameters. A good way to examine the development of chaos is by looking at the Lyapunov exponent  $\lambda$ . The Lyapunov exponent is based on the differences between solutions if we make a small change in the initial conditions. Consider two starting values of  $x$ :  $x_0 + \varepsilon$  and  $x_0$ . The difference between the solutions for these starting conditions is

$$d^n = f^n(\alpha, x_0 + \varepsilon) - f^n(\alpha, x_0) \quad (4.19)$$

It turns out that this difference grows as




Figure 4.21: Bifurcation diagrams for  $f(\alpha, x) = \alpha x(1 - x)^2$  for a starting value of  $x_0 = 0.1$ .




Figure 4.22: Lyapunov exponent as function of  $\alpha$  for the logistic equation  $f(\alpha, x) = \alpha x(1 - x)$ .

$$d^n = \varepsilon e^{n\lambda} \tag{4.20}$$

Combining these equations we see that

$$\lambda = \frac{1}{n} \ln(e^{n\lambda}) = \frac{1}{n} \ln \left[ \frac{f^n(\alpha, x_0 + \varepsilon) - f^n(\alpha, x_0)}{\varepsilon} \right] = \frac{1}{n} \ln \left[ \left. \frac{df^n(\alpha, x)}{dx} \right|_{x_0} \right] \tag{4.21}$$

Using the chain rule for derivations, we can rewrite this equation as

$$\lambda = \frac{1}{n} \ln \left[ \left. \frac{df^n(\alpha, x)}{dx} \right|_{x_0} \right] = \frac{1}{n} \ln \left[ \left. \frac{df(\alpha, x)}{dx} \right|_{x_{n-1}} \frac{df(\alpha, x)}{dx} \Big|_{x_{n-2}} \cdots \frac{df(\alpha, x)}{dx} \Big|_{x_1} \frac{df(\alpha, x)}{dx} \Big|_{x_0} \right] \tag{4.22}$$

If we take the limit of  $n$  to infinity, we get the definition of the Lyapunov exponent

$$\lambda = \lim_{n \rightarrow \infty} \frac{1}{n} \sum_{i=0}^{n-1} \ln \left| \left. \frac{df(\alpha, x_i)}{dx} \right| \right| \tag{4.23}$$

A plot of the Lyapunov exponent as function of  $\alpha$  for the logistic function  $f(\alpha, x) = \alpha x(1 - x)$  is shown in Fig. 4.22.

If the Lyapunov exponent is negative, the solutions will eventually converge. In Fig. 4.22 we can easily identify the region where a stable solution is achieved. Bifurcation (doubling of the number of solutions) will occur when the Lyapunov exponent is zero. Chaos will occur when the Lyapunov exponent is larger than 0.